Odhad stavu s neznámými parametry

- stavový model
- neznámé parametry
- linearizace vzniklého nelineárního modelu

V programu se provádí simulace s nelineárním stavovým modelem.

Pro odhad stavu se stav modelu se rozšíří o neznámé parametry a tím se stane nelineárním. Model je dále linearizován pomocí prvých dvou členů Taylorova rozvoje. Linearizovaný model je využit pro odhad stavu z generovaného vstupu a výstupu. Odhad se provádí pomocí Kalmanova filtru. Kalmanův filtr je realizován procedurou

[x, xf, rx, yp]=Kalman(x, yt, ut, M, N, F, A, B, G, rw, rv, rx)

kde x je přepočítávaný stav, xf je výsledný odhad stavu v daném čase, rx je přepočítávaná kovariance stavu, yp je predikce výstupu, yt a ut jsou data, M, N, F, A, B, G jsou parametry stavového modelu, rw a rv jsou kovariance stavu a šumu ve stavovém modelu.

Použitý model má tvar

a=.6; // unknown model parameter

sM=[a .5 // simulation parametrs

.1 .8];

sN=[.5; .4];

sA=[1 0];

yt(:,t)=sA*x+sqrt(rv)*rand(1,1,'norm'); x=sM*x+sN*ut(:,t)+sqrt(rw)*rand(2,1,'norm');

$$
\begin{aligned}
x_{t+1} &= Mx_t + Nu + w_t \\
y_t &= Ax_t + v_t
\end{aligned}
$$

kde $M = \begin{bmatrix} a & 0.5 \\ 1 & 0 \end{bmatrix}$, $\begin{bmatrix} 0.5 \\ 0.4 \end{bmatrix}$, $A = [1, 0]$. V simulaci je $a = 0.6$ v odhadu je a neznámý parametr.

Předpoklady: Neznáme parametry modelu, linearizace modelu pomocí Taylorova rozvoje

Sci značení: $x_{t-1}/x_t/x_{t+1}$ - x, x_t - xf, (ostatní viz Kalmanův filtr nahoře) Parametry modelu pro simulaci jsou sM, sN a sA.

Úloha: Odhad neznámé veličiny, filtrace zašuměného signálu.

Poznámka

Pro správný start odhadování je důležité správné nastavení kovariančních matic. Matice rw se nastavuje většinou jako diagonální s velkými čísly na diagonále ($10^3 - 10^5$). Kovarianční matice rw a rv by měly odpovídat kovariančním maticím šumů w_t a v_t z modelu. POZOR: nejsou to kovarianční matice stavu a výstupu ale jejich poruch. Na správném nastavení těchto matic velmi záleží celý odhad.

Doporučené experimenty

1. Stav je v tomto případě složen z původních stavových proměnných (které se přirozeně v čase mění) a neznámých parametrů (které jsou konstantní). Vzhledem k tomu je třeba volit rozptyly původních stavů větší, zatímco rozptyly neznámých parametrů by měly být velmi malé. Zkuste experimentálně.
2. Nejzajímavější je měnit simulovanou soustavu (včetně dimenze stavu a výstupu). Tato změna ale vyžaduje novou konstrukci rozšířeného stavového modelu (stav je rozšířen o neznámé parametry) a jeho linearizace.

Program

```
// State estimation (model with unknown parameter)
[u,t,n]=file();                         // find working directory
chdir(dirname(n(1)));                   // set working directory
clear("u","t","n")                      // clear auxiliary data
exec("ScIntro.sce",-1),mode(0)          // intro to sesion

nd=100;                       // length of data
a=.6;                         // unknown model parameter
sM=[a .5                      // simulation parametrs
   .1 .8];
sN=[.5; .4];
sA=[1 0];
rv=.01; rw=.01*eye(2,2);      // noise variances

ut=rand(1,nd,'norm');
yt=zeros(1,nd);
xx=zeros(2,1);                 // initial state
xt=xx;

// SIMULATION
for t=1:nd
  xx     =sM*xx+sN*ut(:,t)+sqrt(rw)*rand(2,1,'norm');
  yt(:,t)=sA*xx+sqrt(rv)*rand(1,1,'norm');
  xt(:,t)=xx;
end
xt(3,:)=a;                    // true parameter

Rv=.01; Rw=.01;               // model covariance matrices
Rx=1e6*eye(3,3);              // covariance matrix of state estimate
N=[.5; .4; 0];                // parameters of model extended by unknown
A=[1 0 0];                    //     parmeter
G=0;
B=0;
x=zeros(3,1);
xe=zeros(3,nd);
```

```
// ESTIMATION
for t=1:nd
  M=[x(3) .5 0                // parameters of thr linearized model
     .1   .8 0
     0     0 1];
  dM=[x(3) .5 x(1)
      .1   .8 0
      0     0 1];
  F=(M-dM)*x;

  // KALMAN FILTER
  [x,Rx,yp]=Kalman(x,yt(t),ut(:,t),dM,N,F,A,B,G,Rw,Rv,Rx);
  xe(:,t)=x;
end

// RESULTS
set(scf(1),'position',[100 50 1200 400])
// figure 1
subplot(131)
plot(xt(1,:),'b','linewidth',2)
plot(xe(1,:),'r','linewidth',2)
legend('state','estimate');
title('First state and its estimate','fontsize',5,'FontName','Times')
xlabel('time','fontsize',4,'FontName','Times')
ylabel('values','fontsize',4,'FontName','Times')

// figure 2
subplot(132)
plot(xt(2,:),'b','linewidth',2)
plot(xe(2,:),'r','linewidth',2)
legend('state','estimate');
title('Second state and its estimate','fontsize',5,'FontName','Times')
xlabel('time','fontsize',4,'FontName','Times')
ylabel('values','fontsize',4,'FontName','Times')

// figure 3
subplot(133)
plot(xt(3,:),'b','linewidth',2)
plot(xe(3,:),'r','linewidth',2)
legend('true parameter','estimate');
title('Estimate of the parameter','fontsize',5,'FontName','Times')
xlabel('time','fontsize',4,'FontName','Times')
ylabel('values','fontsize',4,'FontName','Times')
```